
8 Approximate Value Functions

Up to this point, we have assumed that the value function can be represented as a
table. Tables are useful representations only for small, discrete problems. Problems
with larger state spaces may require an infeasible amount of memory, and the
exactmethods discussed in the previous chaptermay require an infeasible amount
of computation. For such problems, we often have to resort to approximate dynamic
programming, where the solution may not be exact.1 One way to approximate

1 A deeper treatment of this topic is
provided by W.B. Powell, Approxi-
mate Dynamic Programming: Solving
the Curses of Dimensionality, 2nd ed.
Wiley, 2011. Relevant insights can
be drawn from a variety of fields
as discussed by W.B. Powell, Re-
inforcement Learning and Stochastic
Optimization. Wiley, 2022.solutions is to use value function approximation, which is the subject of this chapter.

Wewill discuss different approaches to approximating the value function and how
to incorporate dynamic programming to derive approximately optimal policies.

8.1 Parametric Representations

We will use Uθ(s) to denote our parametric representation of the value function,
where θ is the vector of parameters. There are many ways to represent Uθ(s),
several of which will be mentioned later in this chapter. Assuming that we have
such an approximation, we can extract an action according to

π(s) = arg max
a

(

R(s, a) + γ ∑
s′

T(s′ | s, a)Uθ(s
′)

)

(8.1)

Value function approximations are often used in problems with continuous state
spaces, in which case the summation above may be replaced with an integral. The
integral can be approximated using transition model samples.

An alternative to the computation in equation (8.1) is to approximate the
action value function Q(s, a). If we use Qθ(s, a) to represent our parametric

approximation, we can obtain an action according to

π(s) = arg max
a

Qθ(s, a) (8.2)

This chapter discusses how we can apply dynamic programming at a finite set
of states S = s1:m to arrive at a parametric approximation of the value function over
the full state space. Different schemes can be used to generate this set. If the state
space is relatively low-dimensional, we can define a grid. Another approach is to
use random sampling from the state space. However, some states aremore likely to
be encountered than others and are therefore more important in constructing the
value function.We can bias the sampling towardmore important states by running
simulations with some policy (perhaps initially random), from a plausible set of
initial states.

An iterative approach can be used to enhance our approximation of the value
function at the states in S. We alternate between improving our value estimates at
S through dynamic programming and refitting our approximation at those states.
Algorithm 8.1 provides an implementation where the dynamic programming
step consists of Bellman backups as done in value iteration (see section 7.5). A
similar algorithm can be created for action value approximations Qθ.2

2 Several other categories of ap-
proaches for optimizing value func-
tion approximations are surveyed
by A. Geramifard, T. J. Walsh, S.
Tellex, G. Chowdhary, N. Roy, and
J. P. How, “A Tutorial on Linear
Function Approximators for Dy-
namic Programming and Rein-
forcement Learning,” Foundations
and Trends in Machine Learning,
vol. 6, no. 4, pp. 375–451, 2013.

struct ApproximateValueIteration
Uθ # initial parameterized value function that supports fit!
S # set of discrete states for performing backups
k_max # maximum number of iterations

end

function solve(M::ApproximateValueIteration, 𝒫::MDP)
Uθ, S, k_max = M.Uθ, M.S, M.k_max
for k in 1:k_max

U = [backup(𝒫, Uθ, s) for s in S]
fit!(Uθ, S, U)

end
return ValueFunctionPolicy(𝒫, Uθ)

end

Algorithm 8.1. Approximate value
iteration for an MDP with the
parameterized value function ap-
proximation Uθ. We perform back-
ups (defined in algorithm 7.7) at
the states in S to obtain a vec-
tor of utilities U. We then call
fit!(Uθ, S, U), which modifies
the parametric representation Uθ
to better match the value of the
states in S to the utilities in U. Dif-
ferent parametric approximations
have different implementations for
fit!.

All of the parametric representations discussed in this chapter can be used
with algorithm 8.1. To be used with that algorithm, a representation needs to
support the evaluation of Uθ and the fitting of Uθ to estimates of the utilities at
the points in S.

We can group the parametric representations into two categories. The first
category includes local approximation methods, where θ corresponds to the values
at the states in S. To evaluate Uθ(s) at an arbitrary state s, we take a weighted
sum of the values stored in S. The second category includes global approximation
methods, where θ is not directly related to the values at the states in S. In fact, θ
may have far fewer or even far more components than there are states in S.

Both local approximation and many global approximations can be viewed as a
linear function approximation Uθ(s) = θ

⊤β(s), where methods differ in how they
define the vector function β. In local approximation methods, β(s) determines
how toweight the utilities of the states in S to approximate the utility at state s. The
weights are generally nonnegative and sum to 1. In many global approximation
methods, β(s) is viewed as a set of basis functions that are combined in a linear
fashion to obtain an approximation for an arbitrary s.

We can also approximate the action value function using a linear function,
Qθ(s, a) = θ⊤β(s, a). In the context of local approximations, we can provide
approximations over continuous action spaces by choosing a finite set of actions
A ⊂ A. Our parameter vector θwould then consist of |S| × |A| components, each
corresponding to a state-action value. Our function β(s, a) would return a vector
with the same number of components that specifies how to weight together our
finite set of state-action values to obtain an estimate of the utility associated with
state s and action a.

8.2 Nearest Neighbor

A simple approach to local approximation is to use the value of the state in S

that is the nearest neighbor of s. In order to use this approach, we need a distance
metric (see appendix A.3). We use d(s, s′) to denote the distance between two
states s and s′. The approximate value function is then Uθ(s) = θi, where i =

arg minj∈1:m d(sj, s). Figure 8.1 shows an example of a value function represented
using the nearest neighbor scheme.

Figure 8.1. Approximating the val-
ues of states in a two-dimensional,
continuous state space using the
mean of the utility values of their
k-nearest neighbors according to
Euclidean distance. The resulting
value function is piecewise con-
stant.

We can generalize this approach to average together the values of the k-nearest
neighbors. This approach still results in piecewise constant value functions, but
different values for k can result in better approximations. Figure 8.1 shows exam-
ples of value functions approximated with different values for k. Algorithm 8.2
provides an implementation of this.

mutable struct NearestNeighborValueFunction
k # number of neighbors
d # distance function d(s, s′)
S # set of discrete states
θ # vector of values at states in S

end

function (Uθ::NearestNeighborValueFunction)(s)
dists = [Uθ.d(s,s′) for s′ in Uθ.S]
ind = sortperm(dists)[1:Uθ.k]
return mean(Uθ.θ[i] for i in ind)

end

function fit!(Uθ::NearestNeighborValueFunction, S, U)
Uθ.θ = U
return Uθ

end

Algorithm 8.2. The k-nearest
neighbors method, which approxi-
mates the value of a state s based
on the k closest states in S, as deter-
mined by a distance function d. The
vector θ contains the values of the
states in S. Greater efficiency can be
achieved by using specialized data
structures, such as kd-trees, imple-
mented in NearestNeighbors.jl.

8.3 Kernel Smoothing

Another local approximation method is kernel smoothing, where the utilities of
the states in S are smoothed over the entire state space. This method requires
defining a kernel function k(s, s′) that relates pairs of states s and s′. We generally
want k(s, s′) to be higher for states that are closer together because those values
tell us how to weight together the utilities associated with the states in S. This
method results in the following linear approximation:

Uθ(s) =
m

∑
i=1

θiβi(s) = θ
⊤β(s) (8.3)

where
βi(s) =

k(s, si)

∑
m
j=1 k(s, sj)

(8.4)

Algorithm 8.3 provides an implementation of this.
There are many ways that we can define a kernel function. We can define our

kernel to simply be the inverse of the distance between states:

k(s, s′) = max(d(s, s′), ǫ)−1 (8.5)

where ǫ is a small positive constant in order to avoid dividing by zero when s = s′.
Figure 8.2 shows value approximations using several distance functions. As we
can see, kernel smoothing can result in smooth value function approximations, in
contrast with k-nearest neighbors. Figure 8.3 applies this kernel to a discrete hex
world problem and shows the outcome of a few iterations of approximate value
iteration (algorithm 8.1). Figure 8.4 shows a value function and policy learned
for the mountain car problem (appendix F.4) with a continuous state space.

Figure 8.2. Approximating the val-
ues of states in a two-dimensional
continuous state space by assign-
ing values based on proximity to
several states with known values.
Approximations are constructed
using several distance functions.

Another common kernel is the Gaussian kernel:

k(s, s′) = exp

(

−d(s, s′)2

2σ2

)

(8.6)

where σ controls the degree of smoothing.

mutable struct LocallyWeightedValueFunction
k # kernel function k(s, s′)
S # set of discrete states
θ # vector of values at states in S

end

function (Uθ::LocallyWeightedValueFunction)(s)
w = normalize([Uθ.k(s,s′) for s′ in Uθ.S], 1)
return Uθ.θ ⋅ w

end

function fit!(Uθ::LocallyWeightedValueFunction, S, U)
Uθ.θ = U
return Uθ

end

Algorithm 8.3. Locally weighted
value function approximation de-
fined by a kernel function k and a
vector of utilities θ at states in S.

Initial value function U(1)

Iteration 2

Iteration 3

Figure 8.3. Local approximation
value iteration used to iteratively
improve an approximate value
function on the hex world problem.
The five outlined states are used
to approximate the value function.
The value of the remaining states
are approximated using the dis-
tance function ‖s− s′‖2

2. The result-
ing policy is reasonable but nev-
ertheless suboptimal. Positive re-
ward is shown in blue, and nega-
tive reward is shown in red.

Figure 8.4. A utility function and
policy obtained by learning the ac-
tion values for a finite set of states
(white) in the mountain car prob-
lem using the distance function
‖s− s′‖2 + 0.1.

8.4 Linear Interpolation

Figure 8.5. One-dimensional lin-
ear interpolation produces interpo-
lated values along the line segment
connecting two points.

Linear interpolation is another common approach to local approximation. The
one-dimensional case is straightforward, in which the approximated value for a
state s between two states s1 and s2 is

Uθ(s) = αθ1 + (1− α)θ2 (8.7)

with α = (s2 − s)/(s2 − s1). This case is shown in figures 8.5 and 8.6.

s1 s2s

Weight for θ2: 1− α Weight for θ1: α

Figure 8.6. The weight assigned
to each point in one dimension is
proportional to the length of the
segment on the opposite side of the
interpolation state.

Linear interpolation can be extended to a multidimensional grid. In the two-
dimensional case, called bilinear interpolation, we interpolate among four vertices.
Bilinear interpolation is done through single-dimensional linear interpolation,
once in each axis, requiring the utility of four states at the grid vertices. This
interpolation is shown in figure 8.7.

Uθ(s)θ12 θ34

θ24

θ12 = 1D interpolation between θ1 and θ2 along the vertical axis
θ24 = 1D interpolation between θ2 and θ4 along the horizontal axis
θ13 = 1D interpolation between θ1 and θ3 along the horizontal axis
θ34 = 1D interpolation between θ3 and θ4 along the vertical axis

Uθ(s) =















1D interpolation between θ12 and θ34 along the horizontal axis
or

1D interpolation between θ13 and θ24 along the vertical axis

Figure 8.7. Linear interpolation on
a two-dimensional grid is achieved
through linear interpolation on
each axis in turn, in either order.

Given four vertices with coordinates s1 = [x1, y1], s2 = [x1, y2], s3 = [x2, y1],
and s4 = [x2, y2], and a sample state s = [x, y], the interpolated value is

Uθ(s) = αθ12 + (1− α)θ34 (8.8)

=
x2 − x

x2 − x1
θ12 +

x− x1

x2 − x1
θ34 (8.9)

=
x2 − x

x2 − x1
(αθ1 + (1− α)θ2) +

x− x1

x2 − x1
(αθ3 + (1− α)θ4) (8.10)

=
x2 − x

x2 − x1

(

y2 − y

y2 − y1
θ1 +

y− y1

y2 − y1
θ2

)

+
x− x1

x2 − x1

(

y2 − y

y2 − y1
θ3 +

y− y1

y2 − y1
θ4

)

(8.11)

=
(x2 − x)(y2 − y)

(x2 − x1)(y2 − y1)
θ1 +

(x2 − x)(y− y1)

(x2 − x1)(y2 − y1)
θ2 +

(x− x1)(y2 − y)

(x2 − x1)(y2 − y1)
θ3 +

(x− x1)(y− y1)

(x2 − x1)(y2 − y1)
θ4 (8.12)

The resulting interpolationweighs each vertex according to the area of its opposing
quadrant, as shown in figure 8.8.

Figure 8.8. Linear interpolation on
a two-dimensional grid results in
a contribution of each vertex equal
to the relative area of its opposing
quadrant: Uθ(s) = α1θ1 + α2θ2 +
α3θ3 + α4θ4.

Multilinear interpolation in d dimensions is similarly achieved by linearly inter-
polating along each axis, requiring 2d vertices. Here too, the utility of each vertex
is weighted according to the volume of the opposing hyperrectangle. Multilin-
ear interpolation is implemented in algorithm 8.4. Figure 8.9 demonstrates this
approach on a two-dimensional state space.

Figure 8.9. Two-dimensional linear
interpolation over a 3× 7 grid.

8.5 Simplex Interpolation

Multilinear interpolation can be inefficient in high dimensions. Rather than
weighting the contributions of 2d points, simplex interpolation considers only d + 1

points in the neighborhood of a given state to produce a continuous surface that
matches the known sample points.

We start with a multidimensional grid and divide each cell into d! simplexes,
which are multidimensional generalizations of triangles defined by the convex hull
of d + 1 vertices. This process is known as Coxeter-Freudenthal-Kuhn triangulation,3

3 A.W. Moore, “Simplicial Mesh
Generation with Applications,”
Ph.D. dissertation, Cornell Univer-
sity, 1992.

and it ensures that any two simplexes that share a face will produce equivalent
values across the face, thus producing continuity when interpolating, as shown
in figure 8.10.

mutable struct MultilinearValueFunction
o # position of lower-left corner
δ # vector of widths
θ # vector of values at states in S

end

function (Uθ::MultilinearValueFunction)(s)
o, δ, θ = Uθ.o, Uθ.δ, Uθ.θ
Δ = (s - o)./δ
# Multidimensional index of lower-left cell
i = min.(floor.(Int, Δ) .+ 1, size(θ) .- 1)
vertex_index = similar(i)
d = length(s)
u = 0.0
for vertex in 0:2^d-1

weight = 1.0
for j in 1:d

# Check whether jth bit is set
if vertex & (1 << (j-1)) > 0

vertex_index[j] = i[j] + 1
weight *= Δ[j] - i[j] + 1

else
vertex_index[j] = i[j]
weight *= i[j] - Δ[j]

end
end
u += θ[vertex_index...]*weight

end
return u

end

function fit!(Uθ::MultilinearValueFunction, S, U)
Uθ.θ = U
return Uθ

end

Algorithm 8.4. A method for con-
ducting multilinear interpolation
to estimate the value of state vec-
tor s for known state values θ over
a grid defined by a lower-left ver-
tex o and vector of widths δ. Ver-
tices of the grid can all be writ-
ten o + δ.*i for some nonnega-
tive integral vector i. The package
Interpolations.jl also provides
multilinear and other interpolation
methods.

Figure 8.10. Two-dimensional sim-
plex interpolation over a 3× 7 grid.

To illustrate, suppose that we have translated and scaled the cell containing a
state such that the lowest vertex is 0 and the diagonally opposite vertex is 1. There
is a simplex for each permutation of 1 : d. The simplex given by permutation p is
the set of points x satisfying

0 ≤ xp1
≤ xp2 ≤ · · · ≤ xpd

≤ 1 (8.13)

Figure 8.11 shows the simplexes obtained for the unit cube.

(0, 0, 0)

(1, 1, 1)

Figure 8.11. A triangulation of
a unit cube. Based on figure 2.1
of A.W. Moore, “Simplicial Mesh
Generation with Applications,”
Ph.D. dissertation, Cornell Univer-
sity, 1992.

Simplex interpolation first translates and scales a state vector s to the unit
hypercube of its corresponding cell to obtain s′. It then sorts the entries in s′ to
determine which simplex contains s′. The utility at s′ can then be expressed by a
unique linear combination of the vertices of that simplex.

Example 8.1 provides an example of simplex interpolation. The process is
implemented in algorithm 8.5.

Consider a three-dimensional simplex given by the permutation p = [3, 1, 2]

such that points within the simplex satisfy 0 ≤ x3 ≤ x1 ≤ x2 ≤ 1. This
simplex has vertices (0, 0, 0), (0, 1, 0), (1, 1, 0), and (1, 1, 1).

Any point s belonging to the simplex can thus be expressed by a weighting
of the vertices:

We can determine the values of the last three weights in succession:

w4 = s3 w3 = s1 − w4 w2 = s2 − w3 − w4

We obtain w1 by enforcing that the weights sum to 1.
If s = [0.3, 0.7, 0.2], then the weights are

w4 = 0.2 w3 = 0.1 w2 = 0.4 w1 = 0.3

Example 8.1. Simplex interpola-
tion in three dimensions.

(0, 0, 0)

(0, 1, 0)

(1, 1, 0)

(1, 1, 1)

mutable struct SimplexValueFunction
o # position of lower-left corner
δ # vector of widths
θ # vector of values at states in S

end

function (Uθ::SimplexValueFunction)(s)
Δ = (s - Uθ.o)./Uθ.δ
# Multidimensional index of upper-right cell
i = min.(floor.(Int, Δ) .+ 1, size(Uθ.θ) .- 1) .+ 1
u = 0.0
s′ = (s - (Uθ.o + Uθ.δ.*(i.-2))) ./ Uθ.δ
p = sortperm(s′) # increasing order
w_tot = 0.0
for j in p

w = s′[j] - w_tot
u += w*Uθ.θ[i...]
i[j] -= 1
w_tot += w

end
u += (1 - w_tot)*Uθ.θ[i...]
return u

end

function fit!(Uθ::SimplexValueFunction, S, U)
Uθ.θ = U
return Uθ

end

Algorithm 8.5. A method for con-
ducting simplex interpolation to
estimate the value of state vector
s for known state values θ over a
grid defined by a lower-left vertex
o and a vector of widths δ. Ver-
tices of the grid can all be written
o + δ.*i for some nonnegative in-
tegral vector i. Simplex interpola-
tion is also implemented in the gen-
eral GridInterpolations.jl pack-
age.

8.6 Linear Regression

A simple global approximation approach is linear regression, where Uθ(s) is a
linear combination of basis functions, also commonly referred to as features. These
basis functions are generally a nonlinear function of the state s and are combined
into a vector function β(s) or β(s, a), resulting in the approximations

Uθ(s) = θ
⊤β(s) Qθ(s, a) = θ⊤β(s, a) (8.14)

Although our approximation is linear with respect to the basis functions, the
resulting approximation may be nonlinear with respect to the underlying state
variables. Figure 8.12 illustrates this concept. Example 8.2 provides an exam-
ple of global linear value approximation using polynomial basis functions for
the continuous mountain car problem, resulting in a nonlinear value function
approximation with respect to the state variables.

s

U
θ
(s
)

s s2

U
θ
(s
)

Figure 8.12. Linear regressionwith
nonlinear basis functions is linear
in higher dimensions. Here, poly-
nomial regression can be seen as
linear in a three-dimensional space.
The function exists in the plane
formed from its bases, but it does
not occupy the entire plane be-
cause the terms are not indepen-
dent.

Adding more basis functions generally improves the ability to match the target
utilities at the states in S, but too many basis functions can lead to poor approxi-
mations at other states. Principled methods exist for choosing an appropriate set
of basis functions for our regression model.4 4 See chapter 14 of M. J. Kochender-

fer and T.A. Wheeler, Algorithms
for Optimization. MIT Press, 2019.
or chapter 7 of T. Hastie, R. Tibshi-
rani, and J. Friedman, The Elements
of Statistical Learning: Data Mining,
Inference, and Prediction, 2nd ed.
Springer Series in Statistics, 2001.

Fitting linear models involves determining the vector θ that minimizes the
squared error of the predictions at the states in S = s1:m. If the utilities associated
with those states are denoted as u1:m, then we want to find the θ that minimizes

m

∑
i=1

(Ûθ(si)− ui)
2 =

m

∑
i=1

(θ⊤β(si)− ui)
2 (8.15)

We can approximate the value function for the mountain car problem using
a linear approximation. The problem has a continuous state space with two
dimensions consisting of position x and speed v. Here are the basis functions
up to degree six:

β(s) =

[1,

x, v,

x2, xv, v2,

x3, x2v, xv2, v3,

x4, x3v, x2v2, xv3, v4,

x5, x4v, x3v2, x2v3, xv4, v5,

x6, x5v, x4v2, x3v3, x2v4, xv5, v6]

Here is a plot of an approximate value function fit to state-value pairs
from an expert policy:

Example 8.2. Using a linear ap-
proximation to the mountain car
value function. The choice of ba-
sis functions makes a big differ-
ence. The optimal value function
for the mountain car is nonlinear,
with a spiral shape and discontinu-
ities. Even sixth-degree polynomi-
als do not produce a perfect fit.

The optimal θ can be computed through some simple matrix operations. We
first construct a matrix X where each of the m rows Xi,: contains β(si)

⊤.5 It can 5 For an overview of the mathemat-
ics involved in linear regression as
well as more advanced techniques,
see T. Hastie, R. Tibshirani, and J.
Friedman, The Elements of Statisti-
cal Learning: Data Mining, Inference,
and Prediction, 2nd ed. Springer Se-
ries in Statistics, 2001.

be shown that the value of θ that minimizes the squared error is

θ =
(

X⊤X
)−1

X⊤u1:m = X+u1:m (8.16)

where X+ is the Moore-Penrose pseudoinverse of matrix X. The pseudoinverse is
often implemented by first computing the singular value decomposition, X = UΣU∗.
We then have

X+ = UΣ+U∗ (8.17)
The pseudoinverse of the diagonal matrix Σ is obtained by taking the reciprocal
of each nonzero element of the diagonal and then transposing the result.

Figure 8.13 shows how the utilities of states in S are fit with several basis
function families. Different choices of basis functions result in different errors.

Figure 8.13. Linear regressionwith
different basis function families.Algorithm 8.6 provides an implementation for evaluating and fitting linear

regression models of the value function. Example 8.3 demonstrates this approach
with the mountain car problem.

8.7 Neural Network Regression

Neural network regression relieves us of having to construct an appropriate set of
basis functions as required in linear regression. Instead, a neural network is used
to represent our value function. For a review of neural networks, see appendix D.
The input to the neural network would be the state variables, and the output
would be the utility estimate. The parameters θ would correspond to the weights
in the neural network.

mutable struct LinearRegressionValueFunction
β # basis vector function
θ # vector of parameters

end

function (Uθ::LinearRegressionValueFunction)(s)
return Uθ.β(s) ⋅ Uθ.θ

end

function fit!(Uθ::LinearRegressionValueFunction, S, U)
X = hcat([Uθ.β(s) for s in S]...)'
Uθ.θ = pinv(X)*U
return Uθ

end

Algorithm 8.6. Linear regression
value function approximation, de-
fined by a basis vector function
β and a vector of parameters θ.
The function pinv implements the
psuedoinverse. Julia and other lan-
guages support the backslash opera-
tor, which allows us to write X \ U
in place of pinv(X)*U in the fit!
function.

As discussed in appendix D, we can optimize the network weights to achieve
a particular objective. In the context of approximate dynamic programming, we
wouldwant tominimize the error of our predictions, just as we did in the previous
section. However, minimizing the squared error cannot be done through simple
matrix operations. Instead, we generally have to rely on optimization techniques
such as gradient descent. Fortunately, computing the gradient of neural networks
can be done exactly through straightforward application of the derivative chain
rule.

8.8 Summary

• For large or continuous problems, we can attempt to find approximate policies
represented by parameterized models of the value function.

• The approaches taken in this chapter involve iteratively applying steps of
dynamic programming at a finite set of states and refining our parametric
approximation.

• Local approximation techniques approximate the value function based on the
values of nearby states with known values.

• A variety of local approximation techniques include nearest neighbor, kernel
smoothing, linear interpolation, and simplex interpolation.

• Global approximation techniques include linear regression and neural network
regression.

We can apply linear regression to learn a value function for the mountain car
problem. The optimal value function has the form of a spiral, which can be
difficult to approximate with polynomial basis functions (see example 8.2).
We use Fourier basis functions whose components take the following form:

b0(x) = 1/2

bs,i(x) = sin(2πix/T) for i = 1, 2, . . .

bc,i(x) = cos(2πix/T) for i = 1, 2, . . .

where T is the width of the component’s domain. The multidimensional
Fourier basis functions are all combinations of the one-dimensional compo-
nents across the state-space axes. Herewe use an eighth-order approximation,
so i ranges up to 8. The expert policy is to accelerate in the direction of motion.

Example 8.3. Linear regression
using Fourier bases used to ap-
proximate the value function for
the mountain car problem (ap-
pendix F.4). Value functions (top
row) and resulting policies (bot-
tom row) are shown. The globally
approximated value function is a
poor fit despite using eighth-order
Fourier basis functions. The result-
ing approximate policy is not a
close approximation to the expert
policy. The small time step in the
mountain car problem causes even
small changes in the value function
landscape to affect the policy. Op-
timal utility functions often have
complex geometries that can be dif-
ficult to capture with global basis
functions.

• Nonlinear utility functions can be obtained when using linear regression when
combined with an appropriate selection of nonlinear basis functions.

• Neural network regression relieves us of having to specify basis functions, but
fitting them is more complex and generally requires us to use gradient descent
to tune our parametric approximation of the value function.

8.9 Exercises
Exercise 8.1. The value function approximation methods presented in this chapter have
mostly assumed continuous state spaces. The hex world problem, appendix F.1, is discrete,
but most of its states can be mapped to two-dimensional locations. It does, however,
have an additional terminal state that produces zero reward, which does not have a two-
dimensional location. How can one modify the continuous value function approximation
methods in this chapter to handle such a state?
Solution: The hex world problem has the agent navigate through a two-dimensional hexag-
onal grid. However, the agent can enter a single terminal state from one of several grid
hexes. This single terminal state presents a challenge for value function approximation
methods, which often rely on proximity to infer a state’s value.

While the terminal state could be projected to the same state space as the other states,
perhaps far away, this hack would nevertheless force a form of proximity into the terminal
state’s value calculation. Selecting a single position for a state that should be equidistant
to multiple predecessor states introduces bias.

One alternative is to treat the terminal state as a special case. The kernel function could
be modified to produce infinite distance between the terminal state and any other states.

Another option is to adjust the problem to have a terminal state for every hex that
produces a terminal reward. Each terminal state can be coincident with its predecessor
state, but offset in an additional dimension. This transformation maintains proximity at
the expense of additional states.
Exercise 8.2. A tabular representation is a special case of linear approximate value func-
tions. Show how, for any discrete problem, a tabular representation can be framed as a
linear approximate value function.
Solution: Consider a discrete MDP with m states s1:m and n actions a1:n. A tabular repre-
sentation associates a value with each state or state-action pair. We can recover the same
behavior using a linear approximate value function. We associate an indicator function
with each state or state-action pair, whose value is 1 when the input is the given state or
state-action pair and 0 otherwise:

βi(s) = (s = si) =

{

1 if s = si

0 otherwise

or
βij(s, a) = ((s, a) = (si, aj)) =

{

1 if (s, a) = (si, aj)

0 otherwise

Exercise 8.3. Suppose that we have a problem with continuous state and action spaces
and we would like to construct both a local approximation and a global approximation of
the action value function Q(s, a) = θ⊤β(s, a). For global approximation, we choose the
basis functions

β(s, a) =
[

1, s, a, s2, sa, a2
]

Given a set of 100 states S = s1:100 and a set of five actions A = a1:5, how many parameters
are in θ for a local approximation method? Howmany parameters are in θ for the specified
global approximation method?

Solution: In local approximation methods, the state-action values are the parameters. We
will have |S| × |A| = 100× 5 = 500 parameters in θ. In global approximation methods,
the coefficients of the basis functions are the parameters. Since there are six components
in β(s, a), we will have six parameters in θ.

Exercise 8.4. We are given the states s1 = (4, 5), s2 = (2, 6), and s3 = (−1,−1), and their
corresponding values, U(s1) = 2, U(s2) = 10, and U(s3) = 30. Compute the value at state
s = (1, 2) using 2-nearest neighbor local approximation with an L1 distance metric, with
an L2 distance metric, and with an L∞ distance metric.

Solution: We tabulate the distances from s to the points s′ ∈ S as given here:

s′ ∈ S L1 L2 L∞

s1 = (4, 5) 6
√

18 3

s2 = (2, 6) 5
√

17 4

s3 = (−1,−1) 5
√

13 3

Using the L1 norm, we estimate U(s) = (10+ 30)/2 = 20. Using the L2 norm, we estimate
U(s) = (10 + 30)/2 = 20. Using the L∞ norm, we estimate U(s) = (2 + 30)/2 = 16.

Exercise 8.5. We would like to estimate the value at a state s given the values at a set of
two states S = {s1, s2}. If we want to use local approximation value iteration, which of
the following weighting functions are valid? If they are invalid, how could the weighting
functions be modified to make them valid?
• β(s) = [1, 1]

• β(s) = [1− λ, λ] where λ ∈ [0, 1]

• β(s) =
[

e(s−s1)2
, e(s−s2)2

]

Solution: The first set of weighting functions is not valid, as it violates the constraint
∑i βi(s) = 1. We can modify the weighting functions by normalizing them by their sum:

β(s) =
[

1
1+1 , 1

1+1

]

=
[

1
2 , 1

2

]

The second set of weighting functions is valid. The third set of weighting functions is not
valid, as it violates the constraint ∑i βi(s) = 1. We can modify the weighting functions by
normalizing them by their sum:

β(s) =

[

e(s−s1)
2

e(s−s1)
2
+e(s−s2)

2 , e(s−s2)
2

e(s−s1)
2
+e(s−s2)

2

]

Exercise 8.6. Prove that bilinear interpolation is invariant under (nonzero) linear grid
scaling.

Solution: It is straightforward to show that the interpolated value is invariant to a linear
scaling on one or both axes, such as, Ũθ(s̃) = Uθ(s). We show this by substituting all
x- and y-values by their scaled versions x̃ = βx and ỹ = γy, and showing that the grid
scalings cancel out:

Ũθ(s̃) =
(x̃2 − x̃)(ỹ2 − ỹ)

(x̃2 − x̃1)(ỹ2 − ỹ1)
θ1 +

(x̃2 − x̃)(ỹ− ỹ1)

(x̃2 − x̃1)(ỹ2 − ỹ1)
θ2 +

(x̃− x̃1)(ỹ2 − ỹ)

(x̃2 − x̃1)(ỹ2 − ỹ1)
θ3 +

(x̃− x̃1)(ỹ− ỹ1)

(x̃2 − x̃1)(ỹ2 − ỹ1)
θ4

Ũθ(s̃) =
β(x2 − x)γ(y2 − y)

β(x2 − x1)γ(y2 − y1)
θ1 +

β(x2 − x)γ(y− y1)

β(x2 − x1)γ(y2 − y1)
θ2 +

β(x− x1)γ(y2 − y)

β(x2 − x1)γ(y2 − y1)
θ3 +

β(x− x1)γ(y− y1)

β(x2 − x1)γ(y2 − y1)
θ4

Ũθ(s̃) =
(x2 − x)(y2 − y)

(x2 − x1)(y2 − y1)
θ1 +

(x2 − x)(y− y1)

(x2 − x1)(y2 − y1)
θ2 +

(x− x1)(y2 − y)

(x2 − x1)(y2 − y1)
θ3 +

(x− x1)(y− y1)

(x2 − x1)(y2 − y1)
θ4

Ũθ(s̃) = Uθ(s)

Exercise 8.7. Given the four states s1 = [0, 5], s2 = [0, 25], s3 = [1, 5], and s4 = [1, 25], and
a sample state s = [0.7, 10], generate the interpolant equation Uθ(s) for arbitrary θ.

Solution: The general form for bilinear interpolation is given in equation (8.12) and repro-
duced here. To generate the interpolant, we substitute our values into the equation and
simplify:

Uθ(s) =
(x2 − x)(y2 − y)

(x2 − x1)(y2 − y1)
θ1 +

(x2 − x)(y− y1)

(x2 − x1)(y2 − y1)
θ2 +

(x− x1)(y2 − y)

(x2 − x1)(y2 − y1)
θ3 +

(x− x1)(y− y1)

(x2 − x1)(y2 − y1)
θ4

Uθ(s) =
(1− 0.7)(25− 10)

(1− 0)(25− 5)
θ1 +

(1− 0.7)(10− 5)

(1− 0)(25− 5)
θ2 +

(0.7− 0)(25− 10)

(1− 0)(25− 5)
θ3 +

(0.7− 0)(10− 5)

(1− 0)(25− 5)
θ4

Uθ(s) =
9

40
θ1 +

3

40
θ2 +

21

40
θ3 +

7

40
θ4

Exercise 8.8. Following example 8.1, what are the simplex interpolant weights for a state
s = [0.4, 0.95, 0.6]?

Solution: For the given state s, we have 0 ≤ x1 ≤ x3 ≤ x2 ≤ 1, and so our permutation
vector is p = [1, 3, 2]. The vertices of our simplex can be generated by starting from (0, 0, 0)

and changing each 0 to a 1 in reverse order of the permutation vector. Thus, the vertices of
the simplex are (0, 0, 0), (0, 1, 0), (0, 1, 1), and (1, 1, 1).

Any point s belonging to the simplex can thus be expressed by a weighting of the
vertices:

0

0

0

0

0

0

1

1






+ w4







1

1

We can determine the values of the weights in reverse order, finally solving for w1 by
applying the constraint that the weights must sum to 1. We can then compute the weights
for s = [0.4, 0.95, 0.6]:

w4 = s1 w3 = s3 − w4 w2 = s2 − w3 − w4 w1 = 1− w2 − w3 − w4

w4 = 0.4 w3 = 0.2 w2 = 0.35 w1 = 0.05

